
PROC BGLIMM Overview

Mixed Models

A mixed model contains fixed and random effects.

First consider a normal linear mixed model,

Y = Xβ + Zγ + ε

γ ∼ N(0,G)

ε ∼ N(0,R)

where β is fixed effects and γ is random effects.

G-side matrix, G, is the covariance matrix of the random effects.
R-side matrix, R, is the covariance matrix of the residuals.

PROC BGLIMM Overview

GLMM Models

A generalized linear mixed model (GLMM) consists of the following:

a linear predictor η = Xβ + Zγ
a link function

E[Y |β,γ] = g−1(η) = g−1(Xβ + Zγ)

a response distribution in the exponential family (binary, binomial,
Poisson, normal, gamma, negative binomial, ...).

PROC BGLIMM Overview

Bayesian Approach

The Bayesian paradigm: p(θ|Y) ∝ π(θ) · L(Y;θ)

uses the same likelihood function L(Y;θ)

needs a prior for each set of parameters

β ∼ π(β)

γ ∼ N(0,G)

G ∼ π(G)

R ∼ π(R)

treats every parameter, fixed effect or random effect, as a random
variable.
estimates the joint posterior of p(β,γ,R,G|Y) and infers from the
marginal posteriors.

4 / 62



PROC BGLIMM Overview

Bayesian Approach

The Bayesian paradigm: p(θ|Y) ∝ π(θ) · L(Y;θ)

uses the same likelihood function L(Y;θ)

needs a prior for each set of parameters

β ∼ π(β)

γ ∼ N(0,G)

G ∼ π(G)

R ∼ π(R)

treats every parameter, fixed effect or random effect, as a random
variable.
estimates the joint posterior of p(β,γ,R,G|Y) and infers from the
marginal posteriors.

4 / 62



PROC BGLIMM Overview

Bayesian Approach

The Bayesian paradigm: p(θ|Y) ∝ π(θ) · L(Y;θ)

uses the same likelihood function L(Y;θ)

needs a prior for each set of parameters

β ∼ π(β)

γ ∼ N(0,G)

G ∼ π(G)

R ∼ π(R)

treats every parameter, fixed effect or random effect, as a random
variable.
estimates the joint posterior of p(β,γ,R,G|Y) and infers from the
marginal posteriors.

4 / 62



PROC BGLIMM Overview

Bayesian Approach

The Bayesian paradigm: p(θ|Y) ∝ π(θ) · L(Y;θ)

uses the same likelihood function L(Y;θ)

needs a prior for each set of parameters

β ∼ π(β)

γ ∼ N(0,G)

G ∼ π(G)

R ∼ π(R)

treats every parameter, fixed effect or random effect, as a random
variable.

estimates the joint posterior of p(β,γ,R,G|Y) and infers from the
marginal posteriors.

4 / 62



PROC BGLIMM Overview

Bayesian Approach

The Bayesian paradigm: p(θ|Y) ∝ π(θ) · L(Y;θ)

uses the same likelihood function L(Y;θ)

needs a prior for each set of parameters

β ∼ π(β)

γ ∼ N(0,G)

G ∼ π(G)

R ∼ π(R)

treats every parameter, fixed effect or random effect, as a random
variable.
estimates the joint posterior of p(β,γ,R,G|Y) and infers from the
marginal posteriors.

4 / 62



PROC BGLIMM Overview

Advantages of Bayesian Analysis

1 Bayes theorem is a formula for learning, hence provides updates
information in a natural way.

2 Bayesian inference does not require large sample theory.
3 It estimates any functions of parameters directly, without using the

“plug-in” method.
4 In addition to a point estimate, it can give the probability distribution.
5 It provides a convenient setting for a wide range of complex models,

such as hierarchical models and missing data problems.

PROC BGLIMM Overview

Version Information

PROC BGLIMM is new in SAS/STAT 15.1 (SAS 9.4m6).

To find out your version:

proc product_status;
run;

which produces something like:
...
For SAS/STAT ...

Custom version information: 15.1
...

PROC BGLIMM Overview

Feature: Simple Syntax

Transition from PROC MIXED or PROC GLIMMIX to PROC BGLIMM is
straightforward.

The essential statements:

MODEL statement: specifies the response (Y), fixed effects (X),
response distribution, and link function.
RANDOM statement: specifies the random effects (Z) and the G-side
matrix.
REPEATED statement: specifies the R-side matrix.
ESTIMATE statement: computes linear combination of parameters

PROC BGLIMM Overview

Additional Features of PROC BGLIMM

suite of covariance structures (for both G-side and R-side)

covariance heterogeneity modeling
built-in priors
model comparison via DIC
parallel computing combined with optimal sampling

PROC BGLIMM Overview

PROC BGLIMM Syntax

PROC BGLIMM Statement

This statement includes these commonly used options:

DATA= names the input data set
SEED= random seed for simulation
OUTPOST= creates a data set to contain posterior samples
NBI= number of burn-in iterations
NMC= number of Markov chain iterations, excluding the burn-ins
NTHREADS= number of CPUs to run simulations simultaneously on
STATS= posterior statistics (mean, std, credible intervals,

covariance, correlation, percentiles)
DIAG= convergence diagnostics (autocorrelation, ESS, Geweke,

Heidelberger, MCSE, Raftery, ... )
PLOTS= plotting (trace, autocorrelation, density plots)

PROC BGLIMM Overview

PROC BGLIMM Syntax

MODEL response = fixed-effects / DIST= LINK= ...;

Response
Fixed effects
9 response distributions:

I binary, binomial, exponential, gamma, geometric, inverse Gaussian,
negative binomial, normal, Poisson

8 link functions:
I log, logit, probit, inverse, identity, pow(-2), loglog, complementary

loglog

PROC BGLIMM Overview

Simple Linear Regression with Class Variable

proc bglimm data=Sashelp.Class seed=436792;
class sex;
model Weight = Height Age Sex / dist=normal

coeffprior=normal(var=1e6);
run;

Normal is the default distribution for a continuous response.
COEFFPRIOR= specifies the prior of β. The default is the flat prior.
SCALEPRIOR= specifies the prior of the scale parameter φ. The
default is inverse gamma.

PROC BGLIMM Overview

Poisson Regression

proc bglimm data=LipCancer seed=10571042;
class County;
model Observed = x County / dist=poisson

link=loglog offset=LogN;
run;

The default link for Poisson is the log function.
OFFSET= adds an offset variable to the linear predictor.

PROC BGLIMM Overview

Binomial with Probit Link

proc bglimm data=MultiCenter seed=976352;
class Center Group;
model SideEffect/N = Group / link=probit noint;

run;

The default link for the binomial distribution is the logit function.
NOINT excludes the intercept from the model.

PROC BGLIMM Overview

PROC BGLIMM Syntax

RANDOM random-effects / SUB= GROUP= TYPE= ...;

Defines Z and the G-side matrix.

SUB= identifies the subjects for the random effects. A set of random
effects is estimated for each subject level.
GROUP= specifies groups by which to vary the covariance
parameters; each new level of the grouping effect produces a new set
of covariance parameters.
TYPE= defines the type of the G-side matrix.

I 13 choices: VC, CS, AR, ARMA, TOEP, UN, ...

PROC BGLIMM Overview

Logistic Random-Effects Model

proc bglimm data=MultiCenter seed=976352;
class Center Group;
model SideEffect/N = Group / noint;
random int / sub = Center;

run;

The random effects are assumed to be normally distributed:

γi ∼ N(0,Gi)

PROC BGLIMM Overview

Multiple RANDOM Statements

You can add multiple random effects to the model.

proc bglimm data=a;
class Analyst Run Plate;
model log_assay = Analyst;
random int conc / sub=run(analyst)

covprior=uniform(lower=0, upper=2);
random int conc / sub=plate(run*analyst)

covprior=halfnormal(var=4);
run;

The random effects can be nested or nonnested.

The COVPRIOR= option specifies the prior for the G-side matrix.

PROC BGLIMM Overview

PROC BGLIMM Syntax

REPEATED repeated-effect / SUB= GROUP= TYPE= ...;

Specifies the R-side matrix.

A repeated-effect is required to define the proper location of the
repeated responses.
SUB= groups repeated measures together for the same subject.
GROUP= specifies groups by which to vary the covariance
parameters; each new level of the grouping effect produces a new set
of covariance parameters.
TYPE= defines the type of the R-side matrix.

I 13 choices: VC, CS, AR, ARMA, TOEP, UN, ...

PROC BGLIMM Overview

Repeated Measures Model

proc bglimm data=Fev nmc=10000 seed=44672057
outpost=FevOut;

class Drug Patient Hour;
model FEV = BaseVal Drug Hour;
random int / sub=Patient;
repeated Hour / sub=Patient(Drug) type=un;

run;

Repeated measurements can be balanced or unbalanced.
The REPEATED statement is available only for the normal
distribution with the identity link in SAS/STAT 15.1.

18 / 62

PROC BGLIMM Overview

Model Heterogeneity

The GROUP= option models different covariance matrices for different
groups:

proc bglimm data=pr seed=475193 outpost=pr_out;
class Person Gender Time;
model Distance = Age|Gender;
repeated Time / type=un sub=Person group=Gender;

run;

PROC BGLIMM Overview

ESTIMATE Statement

ESTIMATE "label" estimate-specification < / options>;

This provides a mechanism for computing linear combinations of the
parameters L′θ and θ′ = (β′ γ ′).

Example: To compare the first level with the second level for the effect A:

estimate ’A 1 vs 2’ A 1 -1 0;

Random effects are added after the vertical bar:

estimate ’A 1 vs 2’ A 1 -1 0 | A*Block .25 .25 .25 .25 -.25 -.25 -.25 -.25 / e;

Procedure Details

Sampling

PROC BGLIMM updates parameters conditionally and sequentially through
Gibbs sampling.

1 The fixed-effects parameters are drawn together first at each iteration.
2 The random-effect parameters are updated by subjects.
3 The G-side covariance parameters are then sampled.
4 Last, the R-side covariance parameters are updated.
5 If present, missing response values are treated as parameters and are

thus also sampled.

Procedure Details

Conditional Posteriors

log(p(β|γ, y,R)) = log(π(β)) +
n∑

i=1

log(f(yi|β,γ,R))

log(p(γj |θ, y)) = log(π(γj |G)) +
∑

i∈{j cluster}
log(f(yi|β,γj,R))

log(p(G|γ)) = log(π(G))) +
∑

j

log(π(γj|G))

log(p(R|γ, y,β)) = log(π(R)) +
n∑

i=1

log(f(yi|β,γ,R))

Procedure Details

Sampling Algorithms

Conjugate sampler
I for β,γi in normal distribution when there is conjugacy
I for a covariance matrix with conjugacy

Gamerman algorithm
I a variation of Metropolis sampling, more efficient for GLMMs
I for β,γi when the response distribution is not normal

No-U-Turn (NUTS) algorithm
I Hamiltonian Monte Carlo sampling with autotuning
I for a covariance matrix when there is no conjugacy

Direct sampler
I sample directly from the likelihood distribution for missing responses

Procedure Details

Prior Distribution: β

flat/constant prior: π(β) ∝ 1 (the default)

model y = x / ... coeffprior=constant;

normal prior:
I β ∼ N(0, σ2I)

model y = x / ... coeffprior=normal(var=1e4);

I β ∼ N(µ,Σ) (via a SAS data set)

model y = x / ... coeffprior=normal(input=MyPrior);

Procedure Details

Prior Distribution: Scale Parameter φ

1 Normal: variance
2 Inverse Gaussian: variance
3 Gamma: shape parameter
4 Negative binomial: number of non-events

Inverse gamma prior

model y = x / ... scaleprior=igamma(shape=a, scale=b);

Gamma prior

model y = x / ... scaleprior=gamma(shape=a, iscale=b);

Improper prior: π(φ) ∝ φ−1

model y = x / ... scaleprior=improper;

Procedure Details

Prior Distribution: G-Side Cov

Specify this in the RANDOM statement, for UN, UN(1), VC, and
TOEP(1) types:

random z / subject=id covprior=iwishart(df=a, scale=b);

random z / subject=id covprior=igamma(shape=a, scale=b);

random z / subject=id covprior=uniform(lower=, upper=);

random z / subject=id covprior=halfcauchy(scale=a);

random z / subject=id covprior=halfnormal(var=a);

random z / subject=id covprior=siwishart(df=a, scale=b, var=c);

Procedure Details

Prior Distribution: R-Side Cov

Specify this in the REPEATED statement, for UN, UN(1), VC, and
TOEP(1) types:

repeated Time / subject=id covprior=iwishart(df=a, scale=b);

repeated Time / subject=id covprior=igamma(shape=a, scale=b);

Examples Logistic Regression

Examples Logistic Regression

PROC BGLIMM: Logistic Regression Example

Consider a study of the analgesic effects of treatments on elderly patients
with neuralgia.

Two test treatments and a placebo are compared
The response variable is whether the patient reported pain or not
Covariates include the age and gender of 60 patients and the duration
of complaint before the treatment began

Examples Logistic Regression

The Data
A subset of the data:

Data Neuralgia;
input Treatment $ Sex $ Age Duration Pain $ @@;
datalines;

P F 68 1 No B M 74 16 No P F 67 30 No
P M 66 26 Yes B F 67 28 No B F 77 16 No
A F 71 12 No B F 72 50 No B F 76 9 Yes
...
P M 67 17 Yes B M 70 22 No A M 65 15 No
P F 67 1 Yes A M 67 10 No P F 72 11 Yes
A F 74 1 No B M 80 21 Yes A F 69 3 No
;

Treatment: A, B, P

Sex: F, M

Pain: Yes, No

Examples Logistic Regression

The Model

A logistic regression is considered for this data set:

paini ∼ binary(pi )

pi = logit(β0 + β1 · SexF,i + β2 · TreatmentA,i

+β3 · TreatmentB,i + β4 · SexF,i · TreatmentA,i

+β5 · SexF,i · TreatmentB,i + β6 · Age + β7 · Duration)

where SexF , TreatmentA, and TreatmentB are dummy variables for the
categorical predictors.

You might want to consider a normal prior with large variance as a
noninformative prior distribution on all the regression coefficients:

π(β0, · · · , β7) ∼ normal(0, var = 1e6)

Examples Logistic Regression

Logistic Regression with Fixed Effects

The following statements fit a Bayesian logistic regression model in PROC
BGLIMM:

proc bglimm data=neuralgia seed=7896 nbi=1000 nmc=20000
thin=2 outpost=SampOut;

class Treatment(ref="P") Sex(ref="M") / param = ref;
model Pain = sex|treatment Age Duration / dist=bin

coeffprior=normal(var=1e6);
run;

This code models the probability of no pain (Pain = No). To change
this to model the probability of pain (Pain = Yes), you can specify the
DESCENDING keyword after the response in the MODEL statement.
The REFERENCE coding is used increating the design matrix
(categorical variables).

Examples Logistic Regression

PROC BGLIMM Output

The BGLIMM Procedure

Model Information

Data Set WORK.NEURALGIA

Response Variable Pain

Distribution Binary

Link Function Logit

Fixed Effects Included Yes

Random Effects Included No

Sampling Algorithm Gamerman

Burn-In Size 1000

Simulation Size 20000

Thinning 2

Random Number Seed 7896

Number of Threads 1

Examples Logistic Regression

Posterior Summary Statistics

Posterior Summaries and Intervals

Parameter N Mean
Standard
Deviation

95%
HPD Interval

Intercept 10000 18.3156 6.7527 5.6462 30.3823

Sex F 10000 2.5533 1.3674 -0.0862 5.2098

Treatment A 10000 4.1330 1.4634 0.9898 6.8380

Treatment B 10000 4.5530 1.5439 1.5893 7.7565

Treatment A*Sex F 10000 -0.8679 1.8766 -4.8454 2.4119

Treatment B*Sex F 10000 -0.3217 1.9256 -3.9514 3.1730

Age 10000 -0.3116 0.0996 -0.5147 -0.1420

Duration 10000 0.00673 0.0345 -0.0518 0.0811

Compare to page 41

Examples Logistic Regression

TAD (Trace, Autocorrelation, Density) Plots

Examples Logistic Regression

Examples Logistic Regression

Odds Ratio
In the logistic model, the log odds function, logit(X ), is given by:

logit(X ) ≡ log
(
Pr(Y = 1 | X )

Pr(Y = 0 | X )

)
= β0 + Xβ1

Suppose that you are interested in calculating the ratio of the odds for the
female patients (SexF = 1) to the male patients (SexF = 0). The log of
the odds ratio is the following:

log(ψ) ≡ log(ψ(SexF = 1, SexF = 0))

= logit(SexF = 1)− logit(SexF = 0)

= (β0 + 1× β1)− (β0 + 0× β1)

= β1

It follows that the odds ratio is:

ψ = exp(β1)

Examples Logistic Regression

ESTIMATE Statement
Use the ESTIMATE statement to get the log of the odds ratio between the
female group and male group conditional on treatment A:

proc bglimm data=neuralgia seed=7896 nbi=1000 nmc=20000
thin=2 outpost=SampOut;

class Treatment(ref="P") Sex(ref="M");
model Pain= sex|treatment Age Duration / dist=bin

coeffprior=normal(var=1e6);
estimate "F vs M, at Trt=A" sex 1 -1

treatment*sex 1 -1 0 0 0 0 / e;
run;

sex 1 -1 : estimates the difference between β1 and β2, which under the
default GLM parametrization, is equal to β1

treatment * sex ... : assigns 1 to the interaction where “treatment=A” and
“sex=F”, and -1 to the interaction where “treatment=A” and
“sex=M”

e : requests that the L matrix coefficients be displayed

Examples Logistic Regression

L Matrix Coefficients
Coefficients for F vs M,

at Trt=A

Parameter Row

Intercept

Sex F 1

Sex M -1

Treatment A

Treatment B

Treatment P

Treatment A*Sex F 1

Treatment A*Sex M -1

Treatment B*Sex F

Treatment B*Sex M

Treatment P*Sex F

Treatment P*Sex M

Age

Duration

Examples Logistic Regression

Posterior Summary Statistics (GLM Parametrization)
Posterior Summaries and Intervals

Parameter N Mean
Standard
Deviation

95%
HPD Interval

Intercept 10000 18.3156 6.7527 5.6462 30.3823

Sex F 10000 2.5533 1.3674 -0.0862 5.2098

Sex M 0 . . . .

Treatment A 10000 4.1330 1.4634 0.9898 6.8380

Treatment B 10000 4.5530 1.5439 1.5893 7.7565

Treatment P 0 . . . .

Treatment A*Sex F 10000 -0.8679 1.8766 -4.8454 2.4119

Treatment A*Sex M 0 . . . .

Treatment B*Sex F 10000 -0.3217 1.9256 -3.9514 3.1730

Treatment B*Sex M 0 . . . .

Treatment P*Sex F 0 . . . .

Treatment P*Sex M 0 . . . .

Age 10000 -0.3116 0.0996 -0.5147 -0.1420

Duration 10000 0.00673 0.0345 -0.0518 0.0811

Compare to page 3541 / 62



Examples Logistic Regression

Results from ESTIMATE Statement

The log odds ratio between the female group and male group conditional
on treatment A:

Results from ESTIMATE Statements

Label Mean
Standard
Deviation

95%
HPD Interval

F vs M, at Trt=A 1.6855 1.2913 -0.7977 4.3331

Examples Logistic Regression

Functions of Parameters

Use the DATA step to compute the
odds ratio by taking exponential of the
ESTIMATE statement variable whose
posterior samples are saved in the
SampOut data set.

data ORSamp;
set SampOut;
OddsRatio=exp(F_vs_M__at_Trt_A);

run;

%sumint(data=ORSamp,var=OddsRatio)

Examples Logistic Regression

Built-In Bayesian Macros

%ESS Effective sample sizes
%GEWEKE Geweke diagnostic
%HEIDEL Heidelberger-Welch diagnostic
%MCSE Monte Carlo standard errors
%POSTACF Autocorrelation
%POSTCOR Correlation matrix
%POSTCOV Covariance matrix
%POSTINT Equal-tail and HPD intervals
%POSTSUM Mean, std, and quantiles
%RAFTERY Raftery diagnostic
%SUMINT Mean, std, and HPD intervals
%TADPLOT Trace plot, AC plot, and density plot

Examples Logistic Random-Effects Model

Outline

1 PROC BGLIMM Overview

2 Procedure Details

3 Examples
Logistic Regression
Logistic Random-Effects Model
Repeated Measurements with Heterogeneity

Examples Logistic Random-Effects Model

Logistic Regression with Random Intercepts

Researchers investigated two
medical procedures.

15 centers were randomly
selected.
Patients were assigned to
two treatment groups, A or
B.
N is number of patients who
received a given procedure.
Sideeffect is the number of
patients who reported side
effects.

data MultiCenter;
input Center Group$ N SideEffect @@;
datalines;

1 A 32 14 1 B 33 18
2 A 30 4 2 B 28 8
3 A 23 14 3 B 24 9
4 A 22 7 4 B 22 10
5 A 20 6 5 B 21 12
6 A 19 1 6 B 20 3
7 A 17 2 7 B 17 6
8 A 16 7 8 B 15 9
9 A 13 1 9 B 14 5

10 A 13 3 10 B 13 1
11 A 11 1 11 B 12 2
12 A 10 1 12 B 9 0
13 A 9 2 13 B 9 6
14 A 8 1 14 B 8 1
15 A 7 1 15 B 8 0
;

Examples Logistic Random-Effects Model

Logistic Regression with Random Intercepts

The response is sample proportions of side effects as binomial ratios
The fixed effect is Group
The random effect cluster is Center

proc bglimm data=MultiCenter nmc=10000 thin=2 seed=976352
outpost=CenterOut plots=all;
class Center Group;
model SideEffect/N = Group / noint;
random int / sub = Center;

run;

Examples Logistic Random-Effects Model

PROC BGLIMM Output

The BGLIMM Procedure

Model Information

Data Set WORK.MULTICENTER

Response Variable SideEffect

Distribution Binomial

Link Function Logit

Fixed Effects Included Yes

Random Effects Included Yes

Sampling Algorithm Gamerman, Conjugate

Burn-In Size 500

Simulation Size 10000

Thinning 2

Random Number Seed 976352

Number of Threads 1

Examples Logistic Random-Effects Model

Posterior Summary Statistics

Posterior Summaries and Intervals

Parameter N Mean
Standard
Deviation

95%
HPD Interval

Group A 5000 -1.3895 0.3102 -2.0071 -0.7956

Group B 5000 -0.8839 0.2968 -1.4819 -0.3186

Random Var 5000 0.9184 0.4198 0.3024 1.7515

Group A vs. Group B
‘Random Var’ measures variability of center-level intercepts.
Each center’s intercept can be printed with MONITOR option in the
RANDOM statement.

Examples Logistic Random-Effects Model

TAD Plots (Trace, Auto-correlation, Density plots)

Examples Logistic Random-Effects Model

Functions of Parameters

Example: probability difference
between A and B:

pDiff =
exp(βb)

1 + exp(βb)
− exp(βa)

1 + exp(βa)

data prob;
set CenterOut;
pDiff = logistic(group_b)

- logistic(group_a);
run;

%sumint(data=prob, var=pDiff)

51 / 62



Examples Logistic Random-Effects Model

Functions of Parameters

Example: probability difference
between A and B:

pDiff =
exp(βb)

1 + exp(βb)
− exp(βa)

1 + exp(βa)

data prob;
set CenterOut;
pDiff = logistic(group_b)

- logistic(group_a);
run;

%sumint(data=prob, var=pDiff)

51 / 62



Examples Logistic Random-Effects Model

Functions of Parameters

Example: probability difference
between A and B:

pDiff =
exp(βb)

1 + exp(βb)
− exp(βa)

1 + exp(βa)

data prob;
set CenterOut;
pDiff = logistic(group_b)

- logistic(group_a);
run;

%sumint(data=prob, var=pDiff)

51 / 62



Examples Repeated Measurements with Heterogeneity

Outline

1 PROC BGLIMM Overview

2 Procedure Details

3 Examples
Logistic Regression
Logistic Random-Effects Model

Examples Repeated Measurements with Heterogeneity

Repeated Measurements with Heterogeneity

A two-treatment trial for
patients with rheumatoid
arthritis
67 subjects enrolled
Subjects were followed
up three times
A grip strength
measurement was taken
at each follow-up visit
A baseline grip strength
(in mmHg) was
measured at the start

data GripData;
input Subject Baseline Treat Gender$ Time Grip;
datalines;
26 175 1 M 1 161
26 175 1 M 2 210
26 175 1 M 3 230
27 165 1 M 1 215
27 165 1 M 2 245
27 165 1 M 3 265
...

71 104 2 F 1 107
71 104 2 F 2 .
71 104 2 F 3 .
72 60 2 F 1 60
72 60 2 F 2 55
72 60 2 F 3 58
;

Examples Repeated Measurements with Heterogeneity

Model Details

A reasonable initial model should involve fairly general specifications for
both the mean and the variance-covariance structure (Littell et al. 2006).

The mean includes three-way interactions of the variables
Gender,Treat, and Time and the Baseline measure.
To allow general within-subject heterogeneity, the unstructured type is
used in the R-side covariance matrix.

I An advantage of considering this most general covariance type is to
inspect the estimates for heterogeneous patterns in both the variances
and correlations.

Repeated Measurements using PROC BGLIMM

The fixed effects contain 12 cell means: 2 treatments by 2 genders by
3 visits.
The REPEATED statement specifies that the repeated measurements
be taken over the Time variable and are grouped according to the
Subject variable.

proc bglimm data=GripData seed=475193;
class Subject Treat Gender Time;
model Grip = Gender*Treat*Time Baseline / noint;
repeated Time / sub=Subject type=un r rcorr;
run;

Posterior R-Side Covariance and Correlation Matrices
Requested via the options r and rcorr.

Estimated R Matrix

Row Col 1 Col 2 Col 3

1 604.96 308.00 288.96

2 308.00 950.48 885.65

3 288.96 885.65 1304.71

Estimated R Correlation
Matrix

Row Col 1 Col 2 Col 3

1 1.0000 0.4062 0.3252

2 0.4062 1.0000 0.7953

3 0.3252 0.7953 1.0000

Grip Strength Measurements over Time by Gender

Between-Subject Heterogeneity by Gender

To account for distinct covariance structures of the two gender groups, you
can fit the model by adding the option GROUP=GENDER to the
REPEATED statement:

proc bglimm data=GripData seed=475193;
class Subject Treat Gender Time;
model Grip = Gender*Treat*Time Baseline / noint;
repeated Time / sub=Subject type=un group=Gender r rcorr;

run;

R-Side Covariance Matrices by Gender

The BGLIMM Procedure

Estimated R Matrix

Group Row Col 1 Col 2 Col 3

Gender F 1 300.08 77.2769 95.2165

Gender F 2 77.2769 267.23 195.20

Gender F 3 95.2165 195.20 257.48

Gender M 1 960.37 591.43 528.93

Gender M 2 591.43 1773.63 1710.94

Gender M 3 528.93 1710.94 2504.11

Between-Subject Heterogeneity in Random effects

You can account for more between-subject heterogeneity by adding a
random statement.

proc bglimm data=GripData seed=475193 nmc=20000 thin=4;
class Subject Treat Gender Time;
model Grip = Gender*Treat*Time Baseline / noint;
random int / sub=Subject group=Gender covprior=uniform;
repeated Time / sub=Subject type=un group=Gender r rcorr

covprior=iw(scale=500);
run;

60 / 62



Examples Repeated Measurements with Heterogeneity

Finishing Thoughts

These development represents part of SAS’ continuing effort to add
Bayesian capability to the software.

We are interested in
I establishing knowledge base and providing how-tos to our users who

are interested in Bayesian GLMMs
I identifying key areas where we can make improvements and

enhancements to our procedures, e.g. PROC MCMC, PROC BGLIMM,
etc

More features and improvements on the way.

For More Information

See the 2019 SAS Global Forum paper ’Introducing the BGLIMM
Procedure for Bayesian Generalized Linear Mixed Models’.

PROC BGLIMM requires SAS/STAT 15.1 (SAS 9.4M6). Complete
documentation of the procedure can be found at http://support.sas.
com/documentation/onlinedoc/stat/151/bglimm.pdf.

You can find additional coding examples at
http://support.sas.com/rnd/app/examples/STATexamples.html.

